

```

% % % % % % % % % % % % % % % % % % % % % % % % % % % % % %
%
%   File:      ch4_dhaasanac.dtr
%   Purpose:   A and B form syncretism in Dhaasanac
%   Author:    Dunstan Brown (December 2011)
%   Email:     d.brown@surrey.ac.uk
%   Address:   Surrey Morphology Group,
%              University of Surrey, Guildford GU2 7XH
%   Documentation:
%              Chapter 4, Brown and Hippiisley (2012)
%              Data from Tosco (2001)
%

```

```

% % % % % % % % % % % % % % % % % % % % % % % % % % % % % %
%
% This fragment generates the positive perfect and
% imperfect paradigms for three example verbs, two
% coronals (fúr 'open' and léet 'fall down'), and one
% non-coronal (rok 'grind'), as well as the positive
% imperative singular, which is the stem plus high tone
% (Tosco 2001: 114). The positive imperative singular
% is the citation form and is therefore used for
% labelling the example lexical entries.
%

```

```

% The purpose of this fragment is to demonstrate that
% we can generate different A and B forms for verbs of
% different types.
%

```

```

% The nodes FORMS, CORONALS, NON_CORONALS:
% these form a hierarchy, with the nodes CORONALS and
% NON_CORONALS inheriting from FORMS. FORMS therefore
% makes generalizations about the forms for both types.
% The node FORMS:
% the first equation states that the perfect ends in -i
% (see Tosco 2001: 113). This ending will be preceded
% by the appropriate consonant grade, which will depend
% on whether the form is the A or B form, and whether
% the verb inherits from CORONALS or NON_CORONALS.
%

```

```

FORMS:

```

```

    <form pf> == "<stem initial>" "<form c grade>" -i

```

```

% The next equation at FORMS creates the forms for the
% imperfect. This is done by evaluating the path
% <stem type> for each lexical item. In this fragment
% either a lexical item will have one vowel associated
% with its stem, or two.
%

```

```

    <form impf> == "<form all "<stem type>" >"

```

```

% If the lexical item has one vowel associated with it, %

```

```

% then it will combine that one vowel with the values %
% for <stem initial>. The values for <stem initial> are %
% defined at the VERB node, but their effect is to %
% create an initial stem which has one vowel, if the %
% lexical item has only one vowel, and two vowels, if %
% the lexical item has two vowels. Hence, for items %
% which have only one vowel, this vowel is repeated for %
% <form all> (associated with the imperfect). This %
% gives us a partial implementation of the Bimoraic %
% Filter (Tosco 2001: 125-126), namely by Stem %
% Adjustment. Nasal extension and reduplication are not %
% dealt with here. The evaluable path at the end of the %
% equation deals with vowel harmony. %

```

```

<form all one_vowel> ==
  "<stem initial>" "<stem vowel_1>"
  "<form c grade>" "<v_harm "<stem vowel_1>" >"

```

```

% The equation below concatenates <stem initial>, the %
% appropriate consonantal grade (<form c grade>) and %
% evaluates the final vowel of the stem to determine %
% the vowel of the imperfect. In this case stems with %
% two vowels do not need to be augmented for the %
% imperfect. %

```

```

<form all two_vowels> ==
  "<stem initial>" "<form c grade>"
  "<v_harm "<stem vowel_2>" >".

```

```

% The node CORONALS: %
% this inherits from the node FORMS. %

```

```

CORONALS:
  <> == FORMS

```

```

% In the two equations below the stem final element in %
% the lexical entry is evaluated to determine the A and %
% B forms. %

```

```

<form c grade _A> == <form _A "<stem final>">
<form c grade _B> == <form _B "<stem final>" >

```

```

% The equations below represent the consonant %
% gradation for the B forms of verbs. Consonant %
% gradation changes the final stem consonant for B %
% forms. This means that by default the grade %
% associated with <form _A> will be the same as what is %
% specified for a lexical item as its final element %
% of the stem. %

```

```

    <form _A> == "<stem final>"

% The spirantization-II rule (Tosco 2001: 21) has been %
% approximated here as a rule which entails that the A %
% form of stems ending in -t will be _ð. %

    <form _A _t> == _ð

% The equations below correspond to table 6 in %
% Tosco (2001: 128). As an example, the equation %
% <form _B _r> == _dɖ can be understood as %
% follows: the B form of a stem ending in r has dɖ in %
% place of r. The other equations are interpreted in %
% the same way. %

    <form _B _r> == _dɖ
    <form _B _t> == _t
    <form _B _d> == _dɖ
    <form _B _n> == _nn
    <form _B _s> == _t
    <form _B _l> == _ll.

% The node NON_CORONALS: %
% this inherits from the node FORMS. %

NON_CORONALS:
    <> == FORMS
    <form type> == non_coronal
    <form all one_vowel> ==
        "<stem initial>" "<form c grade>"
        "<v_harm "<stem vowel_1>" >"
    <form all two_vowels> ==
        "<stem initial>" "<form c grade>"
        "<v_harm "<stem vowel_2>" >"

% Non-Coronals drop their last stem consonant when it %
% is followed by a suffix. This means that the %
% non-coronal is dropped in all verbal forms with the %
% exception of the imperative singular, and the a-form %
% of the short past (Tosco 2001: 132). %
%
% The B form of non-coronals affix yy before the %
% paradigmatic vowel (Tosco 2001: 134). We have treated %
% these two phenomena together, by associating the %
% consonantal grade for A forms as truncated, and the %
% consonantal grade for B forms as yy. If the short %
% past were included in this fragment, it would require %
% the first of these statements to be overridden. %

```

```

    <form c grade _A> ==
    <form c grade _B> == _yy.

% The node VERB (example (35) in chapter 4: §4.5)      %
%                                                    %

VERB:
    <> ==
    <syn> == verb

% Example (35) in chapter 4: §4.5                    %
% The path <index> and its extensions associates      %
% the morphosyntax with the appropriate index.      %
% Second person, third person singular feminine, and %
% first person plural are associated with index B. All %
% other person and number combinations are associated %
% with index A by default.                            %

    <index> == _A
    <index 2nd> == _B
    <index 3rd sg fem> == _B
    <index 1st_excl pl> == _B

% Examples (36) and (37) in chapter 4: §4.5.        %
% The forms of the imperfect positive and perfect   %
% positive are determined by evaluating the index   %
% associated with the particular person and number  %
% information. For example, the form of the second  %
% person (of either number) associated with         %
% <mor pos impf> - the morphosyntax of the imperfect %
% positive - will require the form for              %
% <form impf _B>, once the <index> path is evaluated. %
% On the other hand, the form of the first person  %
% singular, for example, will require the form for %
% <form impf _A>.                                    %
%                                                    %
% These indexed forms are then determined by the    %
% lexical item's membership of a particular form class %
% (coronal or non-coronal).                          %

    <mor pos impf> == "<form impf <index> >"
    <mor pos pf> == "<form pf <index> >"

% By default there is only one stem vowel and so the %
% the value for <stem vowel_2> is 'undefined'.      %

    <stem vowel_2> == undefined

% The initial form of the stem <stem initial> depends %

```

```

% on whether there is a second vowel, <stem vowel_2>. %
    <stem initial> == <stem "<stem vowel_2>" >

% As the default is for <stem vowel_2> to be undefined, %
% the stem will consist of a consonant and a vowel. %

    <stem undefined> == "<stem cons_1>" "<stem vowel_1>"

% Otherwise, if <stem vowel_2> is specified lexically, %
% the stem will consist of a consonant and two vowels, %
% together with a value for tone <stem tone>. %

    <stem> ==
        "<stem cons_1>" "<stem vowel_1>"
        "<stem tone>" "<stem vowel_2>"

% However, as positive verbal forms are unaccented %
% (Tosco 2001: 41), the default for stem tone is set to %
% lack of tone. This is done purely because we are %
% dealing with the positive paradigms. But the value %
% could be different, if this analysis were extended. %

    <stem tone> ==

% The form of the positive imperative singular is the %
% the initial stem plus high tone, and the final %
% element of the stem. %

    <mor pos impv sg> == "<stem initial high_tone>"
                        "<stem final>"

% The position of the high tone is determined by %
% evaluating whether the stem has two vowels or not. %

    <stem initial high_tone> ==
        <stem high_tone "<stem vowel_2>" >

% If the vowel for <stem vowel_2> is 'undefined', then %
% the high tone will be associated with the vowel of %
% the monosyllabic stem. %

    <stem high_tone undefined> ==
        "<stem cons_1>" "<stem vowel_1>"
        "<stem tone high_tone>"

% High tone is represented by ´. %
% %

    <stem tone high_tone> == ´

```

```
% The equations below implement the vowel harmony rule %
% for imperfect. If the final vowel of the stem is o, %
% then the affix of the imperfect is o. If the final %
% vowel of the stem is e, then the affix is e. %
% Otherwise, the imperfect affix is a. %
```

```
<v_harm _e> == -e
<v_harm _o> == -o
<v_harm> == -a.
```

```
% % % % % % % % % % % % % % % % % % % % % % % % % % % % % %
%
%           Example verb lexical entries. %
%
%
% % % % % % % % % % % % % % % % % % % % % % % % % % % % % %
```

```
% The verb rók belongs to the class of non-coronals. %
% The fragment generates forms for the positive perfect %
% and imperfect paradigms and positive imperative %
% singular. These have been checked against Tosco %
% (2001: 451). However, the B form of the imperfect %
% generated is as Tosco (2001: 134). Reduplicated %
% variants are not generated by this fragment. %
```

Rók:

```
<> == VERB
<form> == NON_CORONALS
<gloss> == 'grind'
<stem type> == one_vowel
<stem cons_1> == r
<stem vowel_1> == _o
<stem final> == _k.
```

```
% The verb Léet belongs to the class of coronals. The %
% fragment generates forms for the positive perfect and %
% imperfect paradigms and positive imperative singular. %
% These have been checked against Tosco (2001: 432). %
```

Léet:

```
<> == VERB
<form> == CORONALS
<gloss> == 'fall down'
<stem type> == two_vowels
<stem cons_1> == l
<stem vowel_1> == _e
<stem vowel_2> == _e
<stem final> == _t.
```

```
% Example (34) in chapter 4: §4.5 %
% The verb fúr belongs to the class of coronals. The %
% fragment generates forms for the positive perfect and %
% imperfect paradigms and positive imperative singular. %
% These have been checked against Tosco (2001: 383). %
```

Fúr:

```
<> == VERB
<form> == CORONALS
<gloss> == 'open'
<stem type> == one_vowel
<stem cons_1> == f
<stem vowel_1> == _u
<stem final> == _r.
```

```
% The show paths describe the paradigms for the full %
% morphological model, as discussed in relation to (33) %
% in chapter 4. %
```

```
# show
```

```
<mor pos pf 1st sg>
<mor pos pf 2nd sg>
<mor pos pf 3rd sg fem>
<mor pos pf 3rd sg masc>
<mor pos pf 1st_excl pl>
<mor pos pf 1st_incl pl>
<mor pos pf 2nd pl>
<mor pos pf 3rd pl>

<mor pos impf 1st sg>
<mor pos impf 2nd sg>
<mor pos impf 3rd sg fem>
<mor pos impf 3rd sg masc>
<mor pos impf 1st_excl pl>
<mor pos impf 1st_incl pl>
<mor pos impf 2nd pl>
<mor pos impf 3rd pl>

<mor pos impv sg>.
```

```
# hide FORMS VERB CORONALS NON_CORONALS.
```

